

INF2217 – Exercícios 10

References


[CSE 428: Solutions to exercises on Logic Programming and Prolog]

1. Assume given a set of facts of the form `father(name1, name2)` (`name1` is the father of `name2`).

- Define a predicate `brother(X, Y)` which holds iff `X` and `Y` are brothers.
- Define a predicate `cousin(X, Y)` which holds iff `X` and `Y` are cousins.
- Define a predicate `grandson(X, Y)` which holds iff `X` is a grandson of `Y`.
- Define a predicate `descendent(X, Y)` which holds iff `X` is a descendent of `Y`.
- Consider the following genealogical tree:

```
father(a, b).  
father(a, c).  
father(b, d).  
father(b, e).  
father(c, f).
```

whose graphical representation is:



Say which answers, and in which order, are generated by your definitions for the queries

```
?- brother(X, Y).  
?- cousin(X, Y).  
?- grandson(X, Y).  
?- descendent(X, Y).
```

Justify your answers by drawing the SLD-tree (akas Prolog search tree) for each query, at least schematically.

2. Define a predicate `reverse(L,K)` which holds if and only if the list `K` is the reverse of the list `L`.
3. Consider a representation of sets as lists. Define the following predicates:
 - a. `member(X,L)`, which holds iff the element `X` occurs in `L`.
 - b. `subset(L,K)`, which holds iff `L` is a subset of `K`.
 - c. `disjoint(L,K)`, which holds iff `L` and `K` are disjoint (i.e. they have no elements in common).
 - d. `union(L,K,M)`, which holds iff `M` is the union of `L` and `K`.
 - e. `intersection(L,K,M)`, which holds iff `M` is the intersection of `L` and `K`.
 - f. `difference(L,K,M)`, which holds iff `M` is the difference of `L` and `K`.

Note that the solution to these exercises depends on the way you decide to represent sets. There are various possibilities:

- lists with possible repetitions of the same element
- lists without repetitions
- (in case of lists of integers) ordered lists

For each of these representations, give your solution to the above problems.

4. Define a predicate `length(L,N)` which holds iff `N` is the length of the list `L`.
5. Define a predicate `occurrences(X,L,N)` which holds iff the element `X` occurs `N` times in the list `L`.
6. Define a predicate `occurs(L,N,X)` which holds iff `X` is the element occurring in position `N` of the list `L`.
7. Define a predicate `sumlist(L,N)` which, given a list of integers `L`, returns the sum `N` of all the elements of `L`.
8. Define a predicate `add_up_list(L,K)` which, given a list of integers `L`, returns a list of integers in which each element is the sum of all the elements in `L` up to the same position. Example:


```
?- add_up_list([1,2,3,4],K).
   K = [1,3,6,10];
   no
```
9. Define a predicate `quicksort(L,K)` which, given a list of integers `L`, returns an ordered list `K` obtained from `L` with the method of quicksort.
10. Define a predicate `merge(L,K,M)` which, given two ordered lists of integers `L` and `K`, returns an ordered list `M` containing all the elements of `L` and `K`.

11. Consider a representation of binary trees as terms, as follows

```
emptybt           the empty binary tree
consbt(N,T1,T2)  the binary tree with root N and left
                  and right subtrees T1 and T2
```

- a. Define a predicate `preorder(T,L)` which holds iff `L` is the list of nodes produced by the preorder traversal of the binary tree `T`.
 - b. Define a predicate `search_tree(L,T)` which, given a list of integers `L`, returns a balanced search-tree `T` containing the elements of `L`.
12. A directed graph can be represented in Prolog by listing the arcs among the nodes, as a set of facts (clauses with empty body). For instance, the clause

```
arc(a,b).
```

would represent the existence of an arc going from the node `a` to the node `b`.

- a. Define in Prolog a predicate `path(X,Y)` which holds iff there is an (acyclic) path from the node `X` to the node `Y`. Beware of loops: the graph may contain cycles, but your program should avoid looping on them.
- b. Enrich the previous program so to return not only the answer yes/no, but also, in case of success, the actual path represented as a list of nodes. Namely, define a predicate `path(X,Y,P)` which holds iff `P` is an (acyclic) path from the node `X` to the node `Y`. In case there is more than one acyclic path, the program should generate all of them.
- c. Consider the following graph:

```
arc(a,b).
arc(a,c).
arc(b,c).
arc(b,d).
arc(c,d).
```

What answers are given for the goal `?- path(a,d,P)` by your solution to the last point? Justify your answer by drawing the SLD-tree (akas Prolog search tree), at least schematically.

Answers

Problem 1:

1. `brother(X,Y) :- father(Z,X), father(Z,Y), not(X=Y).` % 6
2. `cousin(X,Y) :- father(Z,X), father(W,Y), brother(Z,W).` % 7
3. `grandson(X,Y) :- father(Z,X), father(Y,Z).` % 8
4. `descendent(X,Y) :- father(Y,X).` % 9
5. `descendent(X,Y) :- father(Z,X), descendent(Z,Y).` % 10


6. ?`brother(X,Y).`
7. `X = b Y = c ;`
8. `X = c Y = b ;`
9. `X = d Y = e ;`
10. `X = e Y = d ;`
11. `No`

12. ?`cousin(X,Y).`
13. `X = d Y = f ;`
14. `X = e Y = f ;`
15. `X = f Y = d ;`
16. `X = f Y = e ;`
17. `No`

18. ?`grandson(X,Y).`
19. `X = d Y = a ;`
20. `X = e Y = a ;`
21. `X = f Y = a ;`
22. `No`

23. ?`descendent(X,Y).`
24. `X = b Y = a ;`
25. `X = c Y = a ;`
26. `X = d Y = b ;`
27. `X = e Y = b ;`
28. `X = f Y = c ;`
29. `X = d Y = a ;`
30. `X = e Y = a ;`
31. `X = f Y = a ;`
32. `No`

We draw the SLD-tree for the first query. Abbreviation: fa stands for father.



Problem 2:

Naive, inefficient (quadratic) solution:

```

naive_reverse([], []).
naive_reverse([X|L], K) :- naive_reverse(L, M), append(M, [X], K).
  
```

Fast (linear), tail-recursive solution:

```

fast_reverse(L, K) :- rev_aux(L, K, []).
rev_aux([], K, K).
rev_aux([X|L], K, M) :- rev_aux(L, K, [X|M]).
  
```

Problem 3:

Note that the solution to these exercises depends on the way you decide to represent sets.

There are various possibilities:

- lists with possible repetitions of the same element
- lists without repetitions
- (in case of lists of integers) ordered lists

For each of these representations, give your solution to the above problems.

a) Lists with possible repetitions of the same element:

```
member(X, [X|_]).
member(X, [_|L]) :- member(X, L).

subset([], _).
subset([X|L], K) :- member(X, K), subset(L, K).

disjoint([], _).
disjoint([X|L], K) :- not(member(X, K)), disjoint(L, K).

union(L, K, M) :- append(L, K, M).

intersection([], _, []).
intersection([X|L], K, M) :- not(member(X, K)), intersection(L, K, M).
intersection([X|L], K, [X|M]) :- member(X, K), intersection(L, K, M).

difference([], _, []).
difference([X|L], K, M) :- member(X, K), difference(L, K, M).
difference([X|L], K, [X|M]) :- not(member(X, K)), difference(L, K, M).
```

b) Lists without repetitions. The main difference is in the implementation of the predicate `union`. Since we do not want repetitions, we have to check, for every element of the first list, that it does not appear already in the second list.

```
union([], K, K).
union([X|L], K, [X|M]) :- not(member(X, K)), union(L, K, M).
union([X|L], K, M) :- member(X, K), union(L, K, M).
```

The other predicates can be defined as before (although for some of them we could give a more efficient definition by using the "cut").

c) Ordered lists of integers. The main difference is that, in the implementation of the predicate `union`, we have to ensure that the output list is ordered. This can be done with a predicate `ordered_merge`:

```
union(L,K,M) :- ordered_merge(L,K,M).
ordered_merge([],K,K).
ordered_merge(L,[],L).
ordered_merge([X|L],[Y|K],[X|M]) :- X < Y, ordered_merge(L,[Y|K],M).
ordered_merge([X|L],[Y|K],[Y|M]) :- Y < X, ordered_merge([X|L],K,M).
ordered_merge([X|L],[Y|K],[X|M]) :- X>=Y, ordered_merge(L,K,M).
```

Some of the other predicates can be defined more efficiently, by using the fact that lists are ordered:

Problem 4:

```
length([],0).
length(_|L,N) :- length(L,M), N is M+1.
```

Problem 5

```
occurrences(_, [],0).
occurrences(X,[X|L],N) :- occurrences(X,L,M), N is M+1.
occurrences(X,[Y|L],N) :- not(X=Y), occurrences(X,L,N).
```

Problem 6:

```
occurs(1,[X|_],X).
occurs(N,_|L,X) :- N > 1, M is N-1, occurs(M,L,X).
```

Problem 7:

```
sumlist([],0).
sumlist([X|L],N) :- sumlist(L,M), N is M+X.
```

Problem 8:

```
add_up_list(L,K) :- aux(L,K,0).
aux([],[],_).
aux([X|L],[Y|K],Z) :- Y is Z+X, aux(L,K,Y).
```

Problem 9:

```
quicksort([], []).
quicksort([X|L],K) :- split(X,L,L1,L2),
                      quicksort(L1,K1),
                      quicksort(L2,K2),
                      append(K1,[X|K2],K).

split(_, [], [], []).
split(X,[Y|L],K,[Y|M]) :- X < Y, split(X,L,K,M).
split(X,[Y|L],[Y|K],M) :- X >= Y, split(X,L,K,M).
```

Problem 10:

If we do not allow multiple elements in the resulting list, then the solution is the same as the `ordered_merge` used above for the definition of union. If allow multiple elements in the resulting list, then we can write the program as follows:

```
merge(L, [], L).
merge([], K, K).
merge([X|L], [Y|K], [X|M]) :- X < Y, merge(L, [Y|K], M).
merge([X|L], [Y|K], [Y|M]) :- X >= Y, merge([X|L], K, M).
```

Problem 11:

```
preorder(emptybt, []).
preorder(consbt(N, T1, T2), L) :-
    preorder(T1, L1), preorder(T2, L2), append([N|L1], L2, L).

search_tree(L, T) :-
    quicksort(L, K), length(K, N), construct_tree(K, N, T).

construct_tree([], 0, emptybt).
construct_tree(L, N, consbt(X, T1, T2)) :-
    N1 is N // 2, % integer division
    N2 is N - N1 - 1,
    divide(L, N1, L1, [X|L2]),
    construct_tree(L1, N1, T1),
    construct_tree(L2, N2, T2).

divide(L, 0, [], L).
divide([X|L], N, [X|L1], L2) :- N > 0, N1 is N-1, divide(L, N1, L1, L2).
```

Problem 12:

In order to avoid cycles, we will use an auxiliary predicate with an additional argument, which represents the list of visited nodes.

```
path(X, Y) :- path_aux(X, Y, [X]). % The third argument is the
                                list of visited nodes
path_aux(X, Y, _) :- arc(X, Y).
path_aux(X, Y, L) :- arc(X, Z), not(member(Z, L)), path_aux(Z, Y, [Z|L]).
```

The list of visited nodes, at the end, contains the nodes of the path. The problem is that this list gets built "in reverse", so we should reverse it to obtain the path. Alternatively, we can search for the path starting from the end node, and walking back towards the start node. Here we follow this approach.

```
path(X, Y, P) :- path_aux(X, Y, [Y], P). % P is the path;
path_aux(X, Y, L, [X|L]) :- arc(X, Y). % The third arg is the list of
                                visited nodes.
path_aux(X, Y, L, P) :- arc(Z, Y), not(member(Z, L)),
path_aux(X, Z, [Z|L], P). %
```

The answers are the following

```
?- path(a,d,P).
P = [a, b, d] ;
P = [a, c, d] ;
P = [a, b, c, d] ;
No
```

The SLD-tree is the following. Abbreviation: m stands for member; pa stands for path_aux:

